IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 1 of 10

Appendix IMAC2 - F: Calculations for the bench test

Data acquisition and reduction

All data were taken with a Hewlett-Packard data acquisition system HP75000 and data 

acquisition software. The data acquisition program was written using HPVEE. All data 

were monitored via graphic windows and transferred to an Excel data sheet. Data 

stored in Excel were then transferred to Engineering Equation Solver (EES). The 

performance of the system was calculated in this EES program and converted again to 

an Excel data sheet for drawing graphs of the results as a form of COP, capacity, flow

rate, torque, etc. 

The following is the calculation model in EES code used to calculate the system 

performance for the bench test. The code includes comments and explanations of the 

calculation being made. The variable names refer to Appendix C. These templates 

were based on SAE ARCRP Pahse I and II with the improvements made during the 

ARCRP program.

Calculation model -- EES template

{========================================================================

This program is for SAE IMAC R134 asystem developed 

based on SAE ARCRP Phase I and Phase II for refrigerant R134a in Jan 05

=========================================================================}

{------------------------------------------------------- Procedure AirFlowRate -----------------------------------------

----------------------------------------------------------------------------------------------------------------------------------

This procedure calculates air flow rates and velocities through the nozzles.

---------------------------------------------------------------------------------------------------------------------------------

Calls: none

Called by: main program

---------------------------------------------------------------------------------------------------------------------------------

Inputs:

CDguess nozzle discharge coefficient guess

D nozzle throat diameter, [m]

Tn nozzle temperature, [C]

Pn nozzle entrance pressure, [kPa]

DPn pressure drop across nozzle, [Pa]

Wn humidity ratio at nozzle

Outputs:

Ma_wet wet air mass flow rate, [kg/s]

Ma_dry dry air mass flow rate, [kg/s]

Q_m3 volumetric flow rate, [m^3/s]

Q_scfm volumetric flow rate, [scfm]

Vel air velocity through nozzle, [m/s]

Vn specific volume of air at nozzle, [m^3/kg]

Re Reynolds Number at nozzle

CDnew discharge coefficient corresponding to Reynolds Number

--------------------------------------------------------------------------------------------------------------------------------}

Procedure AirFlowRate (Nozzle$, CDguess, D, Tn, Pn, DPn, Wn : 

 Ma_wet, Ma_dry, Q_m3, Q_scfm, Vel, Vn, Re, CDold)
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 2 of 10

$Common ENN

IF (Nozzle$ = 'e1') AND (ENN < 1.5) THEN 

 "e1--evap nozzle No.1, ENN=2 for high low load while ENN=1for low load"

Ma_wet = 0

Ma_dry = 0

Q_m3 = 0

Q_scfm = 0

Vel = 0

Vn = 0

Re = 0

CDold = 0

ELSE

An = pi * D^2/4 " nozzle throat area, [m^2] "

Vn = VOLUME(AirH2O,T=Tn,P=Pn,w=Wn)/(1+Wn) 

 " air specific volume at exit of nozzle"

rho = 1/Vn " air density at nozzle, [kg/m^3] "

mu = VISCOSITY(AirH2O, T = Tn, P=Pn, w=Wn) " air viscosity at nozzle, [kg/m-sec] "

 CDnew = CDguess

repeat " iterate to find proper discharge coefficient "

CDold=CDnew

Q_m3 = CDold * An * (2 * DPn * Vn)^0.5

Vel = Q_m3/An

Re = rho * Vel * D/mu

CDnew = .9986 - 7.006/Re^.5 + 134.6/Re 

 "discharge coefficient correlation "

until (abs(CDold - CDnew) < .001)

 

Q_scfm = Q_m3/(1.2 * Vn) * convert(m^3/s, ft^3/min)

 Ma_wet = Q_m3/Vn

Ma_dry = Ma_wet/(1+Wn)

ENDIF

END

{-------------------------------------------- Procedure ChillerCapacity ----------------------------------------

This procedure calculates glycol chiller capacity using the EES BrineProp library.

--------------------------------------------------------------------------------------------------------------------------

Calls: BrineProp

Called by: OutdoorChamberBalance

--------------------------------------------------------------------------------------------------------------------------

Inputs:

Mg glycol mass flow rate, [g/s]

Conc glycol concentration, [mass %]

Tgi glycol inlet temperature, [C]

Tgo glycol outlet temperature, [C]

Outputs:

Q_glycol glycol chiller capacity, [kW]

--------------------------------------------------------------------------------------------------------------------------}

Procedure ChillerCapacity (Mg, Conc, Tgi, Tgo : Q_glycol)

Call BrineProp ('SpecHeat', 'EG', Conc, Tgi : Cpgi)

Call BrineProp ('SpecHeat', 'EG', Conc, Tgo : Cpgo)

Cpg_avg = (Cpgi + Cpgo)/2

Q_glycol = Mg/1000 * Cpg_avg * (Tgo - Tgi)

End
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 3 of 10

{------------------------------------------ Procedure SteamCapacity -----------------------------------------

--------------------------------------------------------------------------------------------------------------------------

This procedure calculates steam capacity.

--------------------------------------------------------------------------------------------------------------------------

Calls: none

Called by: IndoorChamberBalance

--------------------------------------------------------------------------------------------------------------------------

Inputs:

Mg glycol mass flow rate, [g/s]

Conc glycol concentration, [mass %]

Tgi glycol inlet temperature, [C]

Tgo glycol outlet temperature, [C]

Outputs:

Q_glycol glycol chiller capacity, [kW]

--------------------------------------------------------------------------------------------------------------------------}

Procedure SteamCapacity (Mw_kgps, Patm, Ts, Tw : Q_steam)

hs = ENTHALPY(Steam_NBS, T = Ts, x = 1) " saturated steam inlet enthalpy, [kJ/kg] "

hw = ENTHALPY(Steam_NBS, T = Tw, P = Patm) " condensate exit enthalpy, [kJ/kg] "

Q_steam = Mw_kgps * (hs - hw)

End

{------------------------------------- Procedure ChamberLosses ---------------------------------------------

--------------------------------------------------------------------------------------------------------------------------

This procedure calculates the transmission losses through the corresponding chamber

walls.

--------------------------------------------------------------------------------------------------------------------------

Calls: none

Called by: main program

--------------------------------------------------------------------------------------------------------------------------

Inputs:

chamber number telling which chamber equation to use: 1->outdoor, 2->indoor

Tic ceiling temperature inside chamber, [C]

Toc ceiling temperature outside chamber, [C]

Tiw wall temperature inside chamber, [C]

Tow wall temperature outside chamber, [C]

Tif floor temperature inside chamber, [C]

Tof floor temperature outside chamber, [C]

Outputs:

Q_trans transmission losses to chamber walls, [kW]

--------------------------------------------------------------------------------------------------------------------------}

Procedure ChamberLosses (chamber, Tic, Toc, Tiw, Tow, Tif, Tof : Q_trans)

" Outdoor Chamber "

If (chamber = 1) Then

Q_trans = (0.6571 * 7.576 * (Tif - Tof) + 0.2197 * 7.576 * (Tic - Toc) 

 + 0.073 * (8.701 + 5.230) * (Tiw - Tow))/1000

EndIf

" Indoor Chamber "

If (chamber = 2) Then

Q_trans = (0.6571 * 6.493 * (Tif - Tof) + 0.2197 * 6.493 * (Tic - Toc) 

 + 0.073 * (8.701 + 4.483) * (Tiw - Tow))/1000

EndIf

End

{------------------------------------------- Procedure OutdoorChamberBalance ----------------------------
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 4 of 10

--------------------------------------------------------------------------------------------------------------------------

This procedure calculates the heat exchanger heat transfer rate using chamber

calorimetry for the outdoor chamber.

--------------------------------------------------------------------------------------------------------------------------

Calls: none

Called by: main program

--------------------------------------------------------------------------------------------------------------------------

Inputs:

Mg glycol mass flow rate, [g/s]

Conc glycol concentration, [mass %]

Tgi glycol inlet temperature, [C]

Tgo glycol outlet temperature, [C]

Tic ceiling temperature inside chamber, [C]

Toc ceiling temperature outside chamber, [C]

Tiw wall temperature inside chamber, [C]

Tow wall temperature outside chamber, [C]

Tif floor temperature inside chamber, [C]

Tof floor temperature outside chamber, [C]

W chamber electrical power 1, [W]

W2 chamber electrical power 2, [W]

Outputs:

Q_hx heat exchanger heat transfer, [kW]

--------------------------------------------------------------------------------------------------------------------------}

Procedure OutdoorChamberBalance (Mg, Conc, Tgi, Tgo, Tic, Toc, Tiw, Tow, Tif, Tof, W, W2 : Q_hx)

chamber = 1

Call ChillerCapacity (Mg, Conc, Tgi, Tgo : Q_glycol)

Call ChamberLosses (chamber, Tic, Toc, Tiw, Tow, Tif, Tof : Q_trans)

Q_hx = Q_glycol + Q_trans - (W + W2)/1000 

End

{----------------------------------------------- Procedure IndoorChamberBalance --------------------------

--------------------------------------------------------------------------------------------------------------------------

This procedure calculates the heat exchanger heat transfer rate for the indoor chamber.

--------------------------------------------------------------------------------------------------------------------------

Calls: none

Called by: main program

--------------------------------------------------------------------------------------------------------------------------

Inputs:

Mw_kgps condensate rate, [kg/s]

P_atm atmospheric pressure, [kPa]

Ts steam inlet temperature, [C]

Tw condensate exit temperature, [C]

Tic ceiling temperature inside chamber, [C]

Toc ceiling temperature outside chamber, [C]

Tiw wall temperature inside chamber, [C]

Tow wall temperature outside chamber, [C]

Tif floor temperature inside chamber, [C]

Tof floor temperature outside chamber, [C]

W chamber electrical power, [W]

Outputs:

Q_hx heat exchanger heat transfer, [kW]

--------------------------------------------------------------------------------------------------------------------------}

Procedure IndoorChamberBalance (Mw_kgps, Patm, Ts, Tw, Tic, Toc, Tiw, Tow, Tif, Tof, W : 

 Q_hx, Mw_kgps2)
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 5 of 10

 $common Dslope2

chamber = 2

Call SteamCapacity (Mw_kgps, Patm, Ts, Tw : Q_steam)

Call ChamberLosses (chamber, Tic, Toc, Tiw, Tow, Tif, Tof : Q_trans)

Mw_kgps2 = Dslope2 * convert(lbm/s, kg/s) " condensation rate2, [kg/s] "

Q_steam2= Mw_kgps2*(ENTHALPY(Steam_NBS, T = Ts, x = 1) 

 - ENTHALPY(Steam_NBS, T = Ts, x = 0)) 

 Q_hx = Q_steam+Q_steam2 + W/1000 - Q_trans

End

{----------------------------------------------- Procedure Efficiency ---------------------------------------------

--------------------------------------------------------------------------------------------------------------------------

This procedure calculates the various compressor efficiencies.

--------------------------------------------------------------------------------------------------------------------------

Calls: none

Called by: main program

--------------------------------------------------------------------------------------------------------------------------

Inputs:

Mr refrigerant mass flow rate in compressor, [g/s]

Tri refrigerant inlet temperature, [C]

Pri refrigerant inlet pressure, [kPa]

Tro refrigerant outlet temperature, [C]

Pro refrigerant outlet pressure, [kPa]

W_comp compressor work, [kW]

Vc compressor speed, [rpm]

V_disp compressor suction volume, [cc]

Outputs:

h_in inlet refrigerant enthalpy, [kJ/kg]

h_out outlet refrigerant enthalpy, [kJ/kg]

eta_c compression efficiency

eta_v volumetric efficiency

--------------------------------------------------------------------------------------------------------------------------}

Procedure Efficiency (Mr, Tri, Pri, Tro, Pro, W_comp, Vc, V_suc : 

 h_in, h_out, eta_isen, eta_mech, eta_comp, eta_v)

h_in = ENTHALPY(R134a, T = Tri, P = Pri)

s_in = ENTROPY(R134a, T = Tri, P = Pri) " inlet refrigerant entropy, [kJ/kg-K] "

 h_out_isen = ENTHALPY(R134a, P = Pro, s = s_in) 

 " isentropic outlet refrigerant enthalpy, [kJ/kg] "

 h_out = ENTHALPY(R134a, T = Tro, P = Pro) 

 " outlet refrigerant entropy, [kJ/kg-K] "

 eta_isen = (Mr/1000) * (h_out_isen - h_in)/W_comp " Isentropic efficiency "

 eta_mech= (Mr/1000) * (h_out - h_in)/W_comp " Mechanical efficiency "

 eta_comp= (h_out_isen - h_in)/(h_out - h_in) " Compression (or indicated) efficiency "

v_in = VOLUME(R134a, T = Tri, P = Pri) " inlet refrigerant specific volume, [m^3/kg] "

Vdot_c = Mr * v_in " refrigerant displacement rate, [L/s] "

eta_v = (Vdot_c/1000)/(V_suc/1e6 * Vc/60) " Volumetric Efficiency "

End

{******************************************************************************************************************

Main Program Section Begins Here

*******************************************************************************************************************}
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 6 of 10

VersionDate = 031113 " Date of Revision "

{These variables may need to be updated depending on the current system status }

V_disp =160 "compressor suction volume, [cc] "

Xoil=0.0 "assumed oil circulation rate" 

{------------------------------------------ Outdoor Chamber Calculations --------------------------------------------------}

" Chamber Humidity "

Wci = HumRat(AirH2O, P =Patm, T=24, R=0.4) "estimated Rh=40%"

Rhci =RELHUM(AirH2O, T=Tcai, P=Patm, w=Wci) " relative humidity at inlet "

" Air Flow Rate Parameters "

Pcn = Patm - DPcd/1000 - DPcn/1000 "air pressure at nozzle exit, [kPa] "

CDc = 0.98 " discharge coefficient guess value "

D_c1 =0.1778 { 0.17792 original value, 7" } " nozzle 1 diameter, [m] "

D_c2 = 0.1778 {0.17792, 7" } " nozzle 2 diameter, [m] "

Tcn = (Tcn1 + Tcn2)/2 " average nozzle temperature, [C] "

" Air Flow Rate Through Nozzles "

Call AirFlowRate('c1', CDc, D_c1, Tcn1, Pcn, DPcn, Wci :

 ma_wet_c1, ma_dry_c1, AFR_m3_c1, AFR_scfm_c1, Vel_c1, Vn_c1, Re_c1, CDc1)

Call AirFlowRate('c2', CDc, D_c2, Tcn2, Pcn, DPcn, Wci : 

 ma_wet_c2, ma_dry_c2, AFR_m3_c2, AFR_scfm_c2, Vel_c2, Vn_c2, Re_c2, CDc2)

" Total Air Flow Rates "

Ma_outdoor_dry = ma_dry_c1 + ma_dry_c2 " dry air mass flow rate, [kg(dry air)/s] " 

Ma_outdoor = ma_wet_c1 + ma_wet_c2 " air mass flow rate, [kg/s] "

AFR_m3_outdoor = AFR_m3_c1 + AFR_m3_c2 " volumetric air flow rate, [m^3/s] "

AFR_scfm_outdoor = AFR_scfm_c1 + AFR_scfm_c2 " volumetric air flow rate, [scfm] "

" Air-Side Energy Balance "

hcai = ENTHALPY(AirH2O, T = Tcai, P = Patm, W = Wci) " moist inlet air enthalpy, [kJ/kg] "

hcan = ENTHALPY(AirH2O, T = Tcn, P = Pcn, W = Wci) " moist nozzle air enthalpy, [kJ/kg] "

Qoutdoor_air = Ma_outdoor_dry* (hcan - hcai) 

" Refrigerant-Side Energy Balance "

Pcri = Pcro + DPcr " inlet pressure (absolute), kPa "

hcri = ENTHALPY(R134a, T = Tcri, P = Pcri) " refrigerant inlet enthalpy, [kJ/kg] , single phase-gas"

hcro = ENTHALPY(R134a, T = Tcro, P = Pcro) 

“ refrigerant exit enthalpy, [kJ/kg], single phase-subcooled liquid "

Qoutdoor_ref = Mr*(1-Xoil) * (hcri - hcro)/1000

 + Mr*Xoil/1000*(2.0499*(Tcri-Tcro)+2.261e-3/2*(Tcri^2-Tcro^2)) 

Tcro_sat = TEMPERATURE(R134a, P = Pcro, x = 0.5)

DTsubcool=Tcro_sat - Tcro

" Chamber Energy Balance "

Call BRINEPROP ('Density', 'EG', Conc, Tgi : Deng*1000)

Call OutdoorChamberBalance (Mg, Conc, Tgi, Tgo, Tcic, Tcoc, Tciw, Tcow, Tcif, Tcof, Wc, Wc2 : 

 Qoutdoor_chamber)

" Average Energy Balance based on Air-side and chamber balances"

Qoutdoor_avg=0.5*(Qoutdoor_air+Qoutdoor_chamber)

" Error Calculations " 
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 7 of 10

ErrOutdoor_air = (Qoutdoor_air - Qoutdoor_chamber)/Qoutdoor_chamber * 100

ErrOutdoor_ref = (Qoutdoor_ref - Qoutdoor_chamber)/Qoutdoor_chamber * 100

{--------------------------------------Indoor Chamber Calculations --------------------------------------------------------}

" For R134a/R152, the first row of thermocouple grid is out of main stream. "

Teao2 = (Teao21+Teao22+Teao23+Teao24)/4

Teao3 = (Teao31+Teao32+Teao33+Teao34)/4

Teao4 = (Teao41+Teao42+Teao43+Teao44)/4

Teao_avg = (Teao2+Teao3+Teao4)/3

" Chamber Humidity "

Rhei = RELHUM(AirH2O, T = Teai, P = Patm, D = Tdpei) " inlet relative humidity "

Rhen = RELHUM(AirH2O, T = Ten, P = Pen, D = Tdpen) " relative humidity after nozzle "

Wei = HUMRAT(AirH2O, T = Teai, P = Patm, D = Tdpei) " inlet humidity ratio "

Wen = HUMRAT(AirH2O, T = Ten, P = Pen, D = Tdpen) " humidity ratio after nozzle "

" Air Flow Rate Parameters "

Pen = Patm - DPea/1000 - DPen/1000 " air pressure at nozzle entrance, [kPa] "

CDe = 0.975 " discharge coefficient guess value "

D_e1 = 0.06342 {0.0635, 2.5", measured value}

" nozzle 1 (North Side, air flow direction) diameter, [m] "

D_e2 = 0.06348 {0.0635, 2.5", measured value}

" nozzle 2 (South Side, airflow direction) diameter, [m] "

Ten = If(ENN, 2, Ten2, (Ten1 + Ten2)/2,0) " average nozzle temperature, [C] "

" Air Flow Rates Through Nozzles "

Call AirFlowRate('e1', CDe, D_e1, Ten1, Pen, DPen, Wen : ma_wet_e1, ma_dry_e1, AFR_m3_e1, 

AFR_scfm_e1, Vel_e1, Vn_e1, Re_e1, CDe1)

Call AirFlowRate('e2', CDe, D_e2, Ten2, Pen, DPen, Wen : ma_wet_e2, ma_dry_e2, AFR_m3_e2, 

AFR_scfm_e2, Vel_e2, Vn_e2, Re_e2, CDe2)

" Total Air Flow Rates "

Ma_indoor_dry = ma_dry_e1 + ma_dry_e2 " total dry air mass flow rate, [kg/s] "

Ma_indoor_wet = ma_wet_e1 + ma_wet_e2 " total wet air mass flow rate, [kg/s] "

AFR_m3_indoor = AFR_m3_e1 + AFR_m3_e2 " total volumetric air flow rate, [m^3/s] "

AFR_scfm_indoor = AFR_scfm_e1 + AFR_scfm_e2 " total volumetric air flow rate, [scfm] "

" Total Air-Side Heat Transfer "

heai = ENTHALPY(AirH2O, T = Teai, P = Patm, W = Wei) " moist inlet air enthalpy, [kJ/kg] "

hean = ENTHALPY(AirH2O, T = Ten, P = Pen, W = Wen) " moist nozzle air enthalpy, [kJ/kg] "

Qindoor_air = Ma_indoor_dry * (heai - hean)+Qleak_duct-Ma_indoor_dry*(WeiWen)*ENTHALPY(Steam_NBS,T=Ten,x=0)

" heat leak through the duct is added to the air side energy balance"

Qleak_duct=4.8*(Teai-(Ten+Teao_avg)/2)/1000 

" heat leak through the duct. 4.8 is the measured UA value"

" Sensible Air-Side Heat Transfer "

heai_encondition = ENTHALPY(AirH2O, T = Teai, P = Patm, W = Wen) 

Qindoor_sensible_psych = Ma_indoor_dry * (heai_encondition - hean)

Qindoor_latent_psych = Qindoor_air - Qindoor_sensible_psych-Qleak_duct

" Latent Air-Side Heat Transfer "

Mw_kgps_calcu=(Wei-Wen)*Ma_indoor_dry

Mw_kgps = Dslope * convert(lbm/s, kg/s) " condensation rate, [kg/s] "

Mw_gps = Mw_kgps * 1000 " condensation rate, [g/s] "
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 8 of 10

Pv_ei=PRESSURE(Steam_NBS,T=Tdpei,x=1)

Pv_en=PRESSURE(Steam_NBS,T=Tdpen,x=1)

hvapour_in=ENTHALPY(Steam_NBS, T =Teai, P=Pv_ei)

hvapour_out = ENTHALPY(Steam_NBS, T = Ten, x=0)

h_fg= hvapour_in - hvapour_out

Qindoor_latent_cond = Mw_kgps * h_fg

Qindoor_sensible_cond = Qindoor_air - Qindoor_latent_cond-Qleak_duct

" Refrigerant-Side Energy Balance "

{Pero, DPer, Tero are measured variables}

Peri = Pero + DPer " inlet pressure (absolute), kPa "

Teri_sat=TEMPERATURE(R134a, P = Peri, x = 0.5) "2 phase inlet"

Tero_sat= TEMPERATURE(R134a, P = Pero, x = 0.5) " 2-phase saturation temperature on Pero, [C] "

DTsuperheat=Tero-Tero_sat 

hxri = ENTHALPY(R134a, T = Txri, P = Pxri) 

" refrigerant inlet enthalpy, [kJ/kg], single phase-subcooled liquid "

heri = hxri

hero=ENTHALPY(R134a, T = Tero, P = Pero)

Qindoor_ref = Mr*(1-Xoil) * (hero - heri)/1000

 + Mr*Xoil/1000*(2.0499*(Tero-Teri_sat)+2.261e-3/2*(Tero^2-Teri_sat^2)) 

" Refrigerator Qualities " 

h_liq_in = ENTHALPY(R134a, P = Peri, x = 0) " saturated liquid enthalpy, [kJ/kg] "

h_vap_in = ENTHALPY(R134a, P = Peri, x = 1) " saturated vapor enthalpy, [kJ/kg] "

x_in = (heri - h_liq_in)/(h_vap_in - h_liq_in) " inlet quality "

h_liq_out = ENTHALPY(R134a, P = Pero, x = 0) " saturated liquid enthalpy, [kJ/kg] "

h_vap_out = ENTHALPY(R134a, P = Pero, x = 1) " saturated vapor enthalpy, [kJ/kg] "

x_out = (hero - h_liq_out)/(h_vap_out - h_liq_out) " exit quality "

" Chamber Energy Balance " 

Call IndoorChamberBalance (Mw_kgps, Patm, Ts, Tw, Teic, Teoc, 

 Teiw, Teow, Teif, Teof, We : Qindoor_chamber,Mw_kgps2)

" Average Energy Balance based on Air-side and chamber balances"

Qindoor_avg=0.5*(Qindoor_air+Qindoor_chamber)

" Error Calculations "

ErrIndoor_air = (Qindoor_air - Qindoor_chamber)/Qindoor_chamber * 100

ErrIndoor_ref = (Qindoor_ref - Qindoor_chamber)/Qindoor_chamber * 100

{----------------------------------------------Compressor Calculations ---------------------------------------------------}

Prcpo = Pxri + DPhr

P_ratio = Prcpo/Prcpi " compression ratio "

" Efficiency Calculations "

Call Efficiency(Mr*(1-Xoil), Trcpi, Prcpi, Trcpo, Prcpo, W_comp, Vc, 

 V_disp : h_compin, h_compout, eta_isen, eta_mech, eta_comp, eta_v) 

" Compressor Shaft Work "

W_comp = (Fc * convert(lbf-in, N-m)) * (Vc * convert(rev/min, rad/s))/1000
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 9 of 10

DT_sup_cp_in=Trcpi-TEMPERATURE(R134a,P=Prcpi,x=0.5) "compressor inlet superheat"

DT_sup_cp_out=Trcpo-TEMPERATURE(R134a,P=Prcpo,x=0.5) "compressor inlet superheat"

{---------------------------------------------------------System Performance ------------------------------------}

DPsl=Pero-Prcpi

COP_indoor_air = Qindoor_air/W_comp

COP_indoor_ref = Qindoor_ref/W_comp

COP_indoor_chamber = Qindoor_chamber/W_comp

COP_indoor_avg=Qindoor_avg/W_comp

{------------------------------------ Measured parameters from Excel data file -------------------------------------}

" Excel Data "

Pcro=xxxx

DPcr= xxxx

DPca= xxxx

DPcn= xxxx

Wc= xxxx

Wc2= xxxx

Mr= xxxx

Denr= xxxx

Mg= xxxx

Deng= xxxx

Vc= xxxx

Fc= xxxx

Tdpei= xxxx

Tdpen= xxxx

DPcd= xxxx

DPer= xxxx

DPea= xxxx

DPen= xxxx

We= xxxx

Patm= xxxx

G= xxxx

G2= xxxx

Pero= xxxx

count= xxxx

Dslope= xxxx

Dslope2= xxxx

EXV= xxxx

Pxri= xxxx

DPhr= xxxx

Prcpi= xxxx

DPlr= xxxx

Tcic= xxxx

Tciw= xxxx

Tcif= xxxx

Tcoc= xxxx

Tcow= xxxx

Tcof= xxxx

Tcri= xxxx

Tcro= xxxx

Tcai= xxxx

Tcao= xxxx
IMAC Team 2 - Improved Efficiency Final Report December 2006

Appendix F: Calculation for Bench Test Page 10 of 10

Tcn1= xxxx

Tcn2= xxxx

Tgi= xxxx

Tgo= xxxx

Trcpi= xxxx

Trcpo= xxxx

Teic= xxxx

Teiw= xxxx

Teif= xxxx

Teoc= xxxx

Teow= xxxx

Teof= xxxx

Tcph= xxxx

Tero= xxxx

Teai= xxxx

Ten1= xxxx

Ten2= xxxx

Ts= xxxx

Tw= xxxx

Txri= xxxx

Teri= xxxx

Tcro_2= xxxx

Teao21= xxxx

Teao22= xxxx

Teao23= xxxx

Teao24= xxxx

Teao31= xxxx

Teao32= xxxx

Teao33= xxxx

Teao34= xxxx

Teao41= xxxx

Teao42= xxxx

Teao43= xxxx

Teao44= xxxx

ENN= xxxx

outdoor= xxxx

indoor= xxxx

a1File$='xxxxxxxxxxx'

a2Time$='xxxxxxxxxx'